Pensieve header: Making table.tex.

```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Talks\\UNC-1610"];
Once[<< KnotTheory`];
tab = Join[{Knot[0, 1] → {0., 𝔼[1, 0, 0, 0]}}, Get["../../2016-09/tab.m"]];
tab /. (K_ → {_, z_}) :⇒ (z1[K] = z);
Ribbons = {Knot[0, 1], Knot[6, 1], Knot[8, 8], Knot[8, 9], Knot[8, 20], Knot[9, 27],
   Knot[9, 41], Knot[9, 46], Knot[10, 3], Knot[10, 22], Knot[10, 35], Knot[10, 42],
   Knot[10, 48], Knot[10, 75], Knot[10, 87], Knot[10, 99], Knot[10, 123],
   Knot[10, 129], Knot[10, 137], Knot[10, 140], Knot[10, 153], Knot[10, 155]};
```

```
Loading KnotTheory` version of September 6, 2014, 13:37:37.2841.
Read more at http://katlas.org/wiki/KnotTheory.
```

```
z1[Knot[3, 1]]
```

$$\mathbb{E}\left[-1+\frac{1}{t}+t,\ 0,\ 0,\right.$$
$$16+\frac{9\,c}{2}+\frac{2}{t^4}-\frac{2\,c}{t^4}-\frac{7}{t^3}+\frac{11\,c}{2\,t^3}+\frac{14}{t^2}-\frac{8\,c}{t^2}-\frac{18}{t}+\frac{4\,c}{t}-10\,t-10\,c\,t+4\,t^2+8\,c\,t^2-t^3-\frac{3\,c\,t^3}{2}-2\,c\,t^4+$$
$$\left.2\,c\,t^5-\frac{c\,t^6}{2}-4\,u\,w+\frac{2\,u\,w}{t^4}-\frac{7\,u\,w}{2\,t^3}+\frac{9\,u\,w}{2\,t^2}+\frac{u\,w}{2\,t}+6\,t\,u\,w-2\,t^2\,u\,w-\frac{1}{2}\,t^3\,u\,w+\frac{3}{2}\,t^4\,u\,w-\frac{1}{2}\,t^5\,u\,w\right]$$

```
al[K_] := z1[K][[1]];
ap[K_] := al[K] /. t^n_. /; n < 0 :⇒ 0;
```

```
al /@ AllKnots[{0, 5}]
```

$$\left\{1,\ -1+\frac{1}{t}+t,\ 3-\frac{1}{t}-t,\ 1+\frac{1}{t^2}-\frac{1}{t}-t+t^2,\ -3+\frac{2}{t}+2\,t\right\}$$

```
ap /@ AllKnots[{0, 5}]
```

$$\left\{1,\ -1+t,\ 3-t,\ 1-t+t^2,\ -3+2\,t\right\}$$

```
e[K_] := Expand[Together[(t z1[K][[4]] /. c | u | w → 0) + al[K]^3 t^2 D[al[K], t]
                         / ((t - 1)^2 al[K]^2)]];
ep[K_] := e[K] /. t^n_. /; n < 0 :⇒ 0;
```

```
Table[K → e[K], {K, AllKnots[{3, 10}]}] >> etab.m
```

```
ep /@ AllKnots[{0, 5}]
```

$$\left\{0,\ t,\ 0,\ 3\,t+2\,t^3,\ -4+5\,t\right\}$$

```
SymmetryType /@ AllKnots[{0, 10}] // Union
```

```
KnotTheory: The symmetry type data known to KnotTheory` is taken from Charles Livingston's http://www.indiana.edu/~knotinfo/.

KnotTheory: Loading precomputed data in IndianaData`.

{, Chiral, FullyAmphicheiral, NegativeAmphicheiral, Reversible}
```

```
KnotLine[K_] := StringReplace["\\rolcell{n_k}{n^t_{k}}{ap}{ep}{G}{U}{R}{C}", {
   "n" → ToString@K[[1]],
   "k" → ToString@K[[2]],
   "t" → If[AlternatingQ[K], "a", "n"],
   "ap" → ToString[ap[K], FormatType → TeXForm],
   "ep" → ToString[ep[K], FormatType → TeXForm],
   "G" → ToString@ThreeGenus@K,
   "U" → ToString@UnknottingNumber@K,
   "R" → If[MemberQ[Ribbons, K], "\\gY", "\\N"],
   "C" → If[MemberQ[
      {FullyAmphicheiral, NegativeAmphicheiral, ""}, SymmetryType@K], "\\oY", "\\N"]
  }]
```

```
KnotLine /@ AllKnots[{0, 6}]
```

KnotTheory: Loading precomputed data in PD4Knots`.

```
{\rolcell{0_1}{0^a_{1}}{1}{0}{0}{0}{\gY}{\oY},
 \rolcell{3_1}{3^a_{1}}{t-1}{t}{1}{1}{\N}{\N},
 \rolcell{4_1}{4^a_{1}}{3-t}{0}{1}{1}{\N}{\oY},
 \rolcell{5_1}{5^a_{1}}{t^2-t+1}{2 t^3+3 t}{2}{2}{\N}{\N},
 \rolcell{5_2}{5^a_{2}}{2 t-3}{5 t-4}{1}{1}{\N}{\N},
 \rolcell{6_1}{6^a_{1}}{5-2 t}{t-4}{1}{1}{\gY}{\N},
 \rolcell{6_2}{6^a_{2}}{-t^2+3 t-3}{t^3-4 t^2+4 t-4}{2}{1}{\N}{\N},
 \rolcell{6_3}{6^a_{3}}{t^2-3 t+5}{0}{2}{1}{\N}{\oY}}
```

```
Make[n_] := StringJoin @@ Table[
    StringJoin[StringJoin @@ Riffle[L, " &\n"], " \\\\\n\\hline\n"],
    {L, Partition[KnotLine /@ AllKnots[{0, n}], UpTo@2]}
   ];
Make[6]
```

KnotTheory: Loading precomputed data in PD4Knots`.

KnotTheory: The 3-genus data known to KnotTheory` is taken from Charles Livingston's http://www.indiana.edu/~knotinfo/.

KnotTheory: Loading precomputed data in IndianaData`.

```
\rolcell{0_1}{0^a_{1}}{1}{0}{0}{0}{\gY}{\oY} &
\rolcell{3_1}{3^a_{1}}{t-1}{t}{1}{1}{\N}{\N} \\
\hline
\rolcell{4_1}{4^a_{1}}{3-t}{0}{1}{1}{\N}{\oY} &
\rolcell{5_1}{5^a_{1}}{t^2-t+1}{2 t^3+3 t}{2}{2}{\N}{\N} \\
\hline
\rolcell{5_2}{5^a_{2}}{2 t-3}{5 t-4}{1}{1}{\N}{\N} &
\rolcell{6_1}{6^a_{1}}{5-2 t}{t-4}{1}{1}{\gY}{\N} \\
\hline
\rolcell{6_2}{6^a_{2}}{-t^2+3 t-3}{t^3-4 t^2+4 t-4}{2}{1}{\N}{\N} &
\rolcell{6_3}{6^a_{3}}{t^2-3 t+5}{0}{2}{1}{\N}{\oY} \\
\hline
```

```
DeleteFile["table.tex"];
WriteString["table.tex", Make[10]];
Close["table.tex"]
```

```
table.tex
```